

ARBEITEN MIT FUNKTIONEN IN EXCEL

1	Einführung	2
1.1	Was ist eine Funktion?	2
1.2	Der Funktionsassistent	3
1.2.1	<i>Die Handhabung des Funktionsassistenten</i>	3
2	Einfache mathematische Funktionen	4
2.1	Die Summenfunktion: SUMME	4
2.1.1	<i>Einführung</i>	4
2.2	Andere mathematische Funktionen	5
2.2.1	<i>Mathematische Funktionen ohne Argumente</i>	6
2.3	Funktionen zum RUNDEN von Zahlen	7
3	Statistische Funktionen	9
3.1	Größter Wert: MAX	9
3.2	Verwandte Funktionen, die nach demselben Prinzip einzugeben sind	10
3.3	Die Funktionen KGRÖSSTE und KKLEINSTE	11
3.4	Die Funktion RANG	12
4	Funktionen zu Datum und Zeit	13
4.1	Die Funktion TAGE360	13
4.2	Weitere Funktionen zu Datum und ZEIT	14
5	Logische Funktionen	14
5.1	Logische Operatoren	14
5.2	Die Funktion WENN	15
5.2.1	<i>Problemstellung</i>	15
5.2.2	<i>Umkehrung der Bedingung</i>	16
5.2.3	<i>Ein weiteres Beispiel</i>	17
5.2.4	<i>Mehr als zwei möglichen Ergebnissen (Verschachtelte WENN-Funktion)</i>	17
5.2.5	<i>Die verschachtelte WENN-Funktion mit dem Funktionsassistenten</i>	19
5.3	Die Funktion ODER	20
5.4	Die Funktion UND	22
5.5	Die Funktion NICHT	22
5.6	Die Funktionen ZÄHLENWENN und SUMMEWENN	23
6	Matrix-Funktionen	24
6.1	Die Funktion SVERWEIS	24
6.1.1	<i>Einige Regeln zur SVERWEIS-Funktion</i>	26
6.2	Die Funktion WVERWEIS	27
6.3	Die Funktion VERWEIS	28
6.4	Die Funktion VERGLEICH	30
6.5	Die Funktion INDEX	31
6.5.1	<i>INDEX und VERGLEICH kombiniert</i>	32
7	Finanzmathematische Funktionen	33
7.1	Die Funktion RMZ	33
7.2	Die Funktionen ZINS, ZW, BW, ZZR	35

1 EINFÜHRUNG

Tabellenkalkulationsprogramme sind in erster Linie dazu gedacht, aufwendige und immer wiederkehrende Berechnungen ausführen zu lassen. Der Benutzer soll dabei nach Möglichkeit nur veränderliche Eingabe-Daten, meist Zahlen, eingeben. Die Berechnungen sollen automatisch erfolgen.

Um diesem Anspruch gerecht zu werden, müssen jedoch zunächst einmal entsprechende Formulare zur Eingabe der Daten erstellt werden. Hierbei treten immer wieder Probleme auf, die sich scheinbar nicht mit Hilfe von Formeln lösen lassen.

Dabei stellt z. B. das Programm EXCEL eine Fülle von FUNKTIONEN zur Verfügung. Man muss diese nur richtig anzuwenden wissen. Dann lassen sich die meisten Probleme damit lösen.

1.1 Was ist eine Funktion?

Eine FORMEL in EXCEL kann aus den vier Grundrechenarten, eventuell ergänzt durch Klammern, bestehen. Zum Beispiel können Sie zur Berechnung einer Summe die Formel:

```
= B3 + B4 + B5 + B6
```

verwenden.

Zur Berechnung von prozentualen Veränderungen könnte eine Formel so aussehen:

```
= B3 / $B$10 * 100 - 100
```

(Was bedeuteten noch die Dollar-Zeichen? Diese Zelladresse ändert sich nicht mein Auto-Ausfüllen!)

Zur Berechnung einer Summe kann man aber auch die *Summen-Funktion* verwenden, die vor allem bei vielen Zahlenwerten die Eingabe vereinfacht. Diese Funktion hat folgendes Aussehen:

```
= SUMME(B3:B6)
```

Eine Funktion ist generell folgendermaßen aufgebaut:

1. nach dem „=“-Zeichen steht der Name der Funktion
2. direkt anschließend (ohne Leerzeichen) eine „Klammer-auf“.
3. Nach der Klammer folgen sogenannte ARGUMENTE. Dies können vor allem sein:
 - **Zahlen** oder **Texte**
 - **Zelladressen** (z. B. **A1**, **C5**, **\$D\$4**, etc.)
 - **Bereiche von Zelladressen (Vektoren)**, z. B. **A1:A9**, **\$B\$2:\$F\$2**, **\$C2:\$C7** und **Matrizen**, z. B. **C2:E10**, **\$C\$2:\$F\$16**)
 - **Formeln** sowie
 - **Funktionen**.

Werden bei einer Funktion mehrere Argumente angegeben, so müssen diese durch Semikoli (;) voneinander abgegrenzt werden), z. B.:

```
= SUMME(B1:B3;B5:B10;C4)
```

am Ende wird die Klammer wieder geschlossen. Bei Funktionen, die nur ein Argument enthalten, macht dies das Programm auch selbst.

1.2 Der Funktionsassistent

Manche Funktionen in EXCEL sind so aufwendig gestaltet, dass eine Hilfestellung bei der Eingabe sinnvoll ist. Hierfür steht der FUNKTIONSSASSISTENT zur Verfügung.

Dieser Assistent ist vor allem hilfreich bei Funktionen, die viele Argumente enthalten. Einfache Funktionen wie MAX oder MIN sind allerdings schneller eingetippt.

1.2.1 Die Handhabung des Funktionsassistenten

Um eine Funktion mit Hilfe des Assistenten einzugeben, müssen Sie zunächst die Zelle markieren, in der das Ergebnis der Berechnung erscheinen soll.

Anschließend klicken Sie auf das Symbol .

Auf Ihrem Monitor erscheint das folgende Dialogfenster:




Abbildung 1

Meistens werden Sie die gewünschte Funktion unter den „zuletzt verwendeten“ finden. Ansonsten ist folgendes Vorgehen anzuraten:

- ☞ Klicken Sie im linken Menu-Fenster auf „Alle“.
- ☞ Klicken Sie anschließend ins rechte Menu-Fenster (das mit den Funktionen).
- ☞ Tippen Sie den Anfangsbuchstaben der gewünschten Funktion ein und wandern Sie, falls nötig, mit Hilfe der Bildlaufleiste oder mit der Taste „Pfeil nach unten“ noch ein Stückchen nach unten in der Liste.
- ☞ Wenn Sie den Namen der gewünschte Funktion gefunden haben, machen Sie einen Doppelklick darauf.

Im folgenden Dialogfenster können Sie die Argumente eingeben. Dies kann am einfachsten dadurch geschehen, indem Sie die betreffenden Zellen markieren.

Wie man den Funktions-Assistenten bedient, wird weiter unten am Beispiel der Funktion MAX verdeutlicht.

2 EINFACHE MATHEMATISCHE FUNKTIONEN

2.1 Die Summenfunktion: SUMME

Die Summenfunktion wird an dieser Stelle etwas ausführlicher beschrieben. Viele Dinge sind auf andere Funktionen - besonders auf die in diesem Kapitel - anwendbar.

2.1.1 Einführung

Eine der einfachsten Funktionen in Excel ist die Summenfunktion. Da diese außerdem sehr häufig genutzt wird, gibt es hierfür ein Schnell-Symbol.

Zur Erläuterung geben Sie die folgende Tabelle ein und markieren Sie die Zelle B6.

	A	B
1	Artikel	Menge
2	Artikel 1	100
3	Artikel 2	120
4	Artikel 3	150
5	Artikel 4	300
6	Summe	

Abbildung 2

Ohne Verwendung einer Funktion könnten Sie folgende Formel eingeben:

= B2 + B3 + B4 + B5

Diese Formel liefert das korrekte Ergebnis. Bei längeren Datenreihen bietet sich jedoch eine einfachere Formel an.

Markieren Sie die Zelle B6 und klicken Sie auf das Symbol für die Summe .

Sie werden sehen, dass das Programm die Zellen B2 bis B5 markiert. Die Markierung besteht aus einer blinkenden, unterbrochenen Umrahmung. Gleichzeitig zeigen die *Bearbeitungsleiste* und die Zelle B6 die Formel

= SUMME(B2:B5)

an.

Diese Formel ist korrekt. Durch nochmaliges Klicken auf das Summen-Symbol oder durch <Enter> wird die Eingabe abgeschlossen. In der Zelle B6 steht nun das Ergebnis der Berechnung: 670.

Um die Summe zu berechnen, gibt es eine weitere Möglichkeit. Markieren Sie die Zellen B2 bis B5 oder B2 bis B6, klicken Sie anschließend auf das Summen-Symbol. In der Zelle B6 erscheint das Ergebnis.

Beide Methoden haben Ihre Vorteile. Geben Sie als weiteres Beispiel die folgende Tabelle ein:

	A	B	C	D
1		Produkt 1	Produkt 2	Summe
2	<i>Jan</i>	100	1000	
3	<i>Feb</i>	150	900	
4	<i>Mrz</i>	125	950	
5	Summe			

Abbildung 3

Es sollen nun die Summen der Monatsumsätze und die Summen der monatlichen Umsätze beider Produkte sowie die Gesamtsumme berechnet werden.

Markieren Sie dazu die komplette Tabelle, von Zelle A1 bis D5 und klicken Sie das Summen-Symbol an. Es werden sofort alle Summen korrekt berechnet. Die Tabelle sollte dann so aussehen:

	A	B	C	D
1		Produkt 1	Produkt 2	Summe
2	<i>Jan</i>	100	1000	1100
3	<i>Feb</i>	150	900	1050
4	<i>Mrz</i>	125	950	1075
5	Summe	375	2850	3225

Abbildung 4

Wenn Sie die Zahlen überprüfen, werden Sie feststellen, dass alles korrekt berechnet wurde.

2.2 Andere mathematische Funktionen

Es gibt eine ganze Reihe weiterer Funktionen, die unter dieser Kategorie zusammengefasst werden. Dazu gehören z. B.:

- =WURZEL(ZAHL)** Berechnet die Quadratwurzel.
- =LOG10(ZAHL)** Berechnet den Zehner-Logarithmus der Zahl
- =FAKULTÄT(ZAHL)** Berechnet die Fakultät (=1*2*3*...*Zahl)
- =GANZZAHL(ZAHL)** schneidet alle Nachkomma-Stellen der berechneten Zahl ab.
- =ABS(ZAHL)** Liefert den „Absolutwert“ (Das Vorzeichen fällt weg).
- =RÖMISCH(ZAHL)** Wandelt eine ganze Zahl in eine Römische Zahl um, z. B.:
3 ergibt III, 4 ergibt IV.
- =PRODUKT(ZAHL1;ZAHL2;...)** liefert analog zur Summenfunktion das Produkt aller Zahlen. Man könnte dafür auch schreiben:
=Zahl1*Zahl2*Zahl3...
- =POTENZ(ZAHL;EXPONENT)** Potenziert die Zahl mit dem angegebenen Exponenten, z. B.: =Potenz(4;2) ergibt 16, da $4^2=16$. Statt der vier eine Zelleadresse angeben.
- =GGT(ZAHL1, ZAHL2)** liefert den **größten gemeinsamen Teiler** der beiden Zahlen. Beispiel: der größte gemeinsame Teiler von 18 und 30 ist 6, da

sich beide Zahlen glatt durch sechs teilen lassen. Auch 2 und 3 sind gemeinsame Teiler, die 6 ist aber eben der größte!

=KGV(ZAHL1, ZAHL2) kennt man wie GGT sicher auch noch aus der Schule: Es wird das **kleinste gemeinsame Vielfache** bestimmt. Im einfachsten Fall ist das dasselbe wie das Produkt beider Zahlen, es gibt aber auch KGVs, die kleiner sind. Beispiel: das KGV von 4 und 6 ist 12, das sich die 12 durch beide Zahlen glatt teilen lässt.

=QUOTIENT(ZAHL1, ZAHL2) der **Quotient** ist das **Ergebnis einer Division**. Hier wird jedoch nur die **ganze Zahl** – also ohne Nachkommastellen – ausgegeben. Ergänzend dazu gibt die Funktion

=REST(ZAHL1, ZAHL2) den verbleibenden **Rest einer Division** aus.
BEISPIEL: *7 geteilt durch 3 ergibt 2 Rest 1* (so haben wir das mal in der Grundschule gelernt, ab der vierten Klasse kam dabei 2,33333 raus). Die Funktion **=QUOTIENT(7;3)** ergibt also 2, **=REST(7;3)** ergibt 1.

=KOMBINATIONEN(N;K) Der Mathematiker kennt diese Funktion unter dem Begriff „**N über K**“. Berechnet wird, wie viele mögliche Kombinationen es dafür gibt, *K* Zahlen aus insgesamt *N* Zahlen auszuwählen, wobei es nicht auf die Reihenfolge der *K* Zahlen ankommen soll. Z. B. : Wie viele Möglichkeiten gibt es, sechs richtige Lottozahlen (aus 49) zu tippen. Die Funktion **=Kombinationen(49;6)** liefert als Ergebnis knapp 14 Millionen (13.983.816).

=SIN(ZAHL), COS(ZAHL), TAN(ZAHL) die Funktionen berechnen den **Sinus**, **Kosinus** oder **Tanges** eines Winkels. Daneben gibt es noch eine Reihe anderer trigonometrischer Funktionen.

2.2.1 Mathematische Funktionen ohne Argumente

Es gibt in Excel eine ganze Reihe von Funktionen, die ohne die Angabe von Argumenten (Zahlen, Zelladressen etc.) innerhalb der Klammern auskommen. Die beiden Klammern müssen bei diesen Funktionen mit angegeben werden!

Aus dem Bereich mathematischer Funktionen einige Beispiele:

=PI() Liefert als Ergebnis die Zahl Pi (=3,14....)

=ZUFALLSZAHL() Gibt als Ergebnis irgendeine Dezimalzahl zwischen 0 und 1, die nach einem Zufallsprinzip ermittelt wird.

2.3 Funktionen zum RUNDEN von Zahlen

Geben Sie die folgende Tabelle ein und berechnen Sie die Werte in den Spalten „Umsatzsteuer“ und „Bruttopreis“ mit entsprechenden Formeln:

In Zelle C3 muss stehen: **=B3*\$C\$2**
 In Zelle D3: **=SUMME(C3:D3)**
 und in D7: **=SUMME(D3:D6)**

	A	B	C	D
1		Netto- preis	Umsatz- steuer	Brutto- preis
2			16%	
3	Artikel 1	1,95 DM	0,31 DM	2,26 DM
4	Artikel 2	2,95 DM	0,47 DM	3,42 DM
5	Artikel 3	3,95 DM	0,63 DM	4,58 DM
6	Artikel 4	4,95 DM	0,79 DM	5,74 DM
7	Rechnungsbetrag			16,01 DM

Abbildung 5

Wenn Sie den Rechnungsbetrag mit Hilfe Ihres Taschenrechners (oder im Kopf?) kontrollieren, entdecken Sie einen Fehler. Das korrekte Ergebnis müsste lauten: 16,00 DM! Woher kommt der eine Pfennig?

Erklärung: Ergänzen Sie in Spalte C und D je eine Nachkommastelle mit Hilfe des Symbols „Dezimalstelle hinzufügen“. Die Tabelle sieht dann so aus:

	A	B	C	D
1		Netto- preis	Umsatz- steuer	Brutto- preis
2			16%	
3	Artikel 1	1,95 DM	0,312 DM	2,262 DM
4	Artikel 2	2,95 DM	0,472 DM	3,422 DM
5	Artikel 3	3,95 DM	0,632 DM	4,582 DM
6	Artikel 4	4,95 DM	0,792 DM	5,742 DM
7	Rechnungsbetrag			16,008 DM

Abbildung 6

Wenn Sie die Ergebnisse jetzt kontrollieren, stimmt der Rechnungsbetrag. Das Programm Excel hat auch zuvor mit diesen Zahlen gerechnet, die dritte Dezimalstelle wurde lediglich nicht angezeigt. Daher wurden bei der Umsatzsteuer und beim Bruttopreis die Zahlen abgerundet. Beim Rechnungsbetrag jedoch – mathematisch korrekt – aufgerundet. Nun fallen solche Rundungsdifferenzen häufig nicht ins Gewicht. In dem Beispiel müsste man jedoch damit rechnen, dass ein Kunde diese Rechnung reklamiert.

Daher ist es in vielen Fällen gewünscht, Zwischenergebnisse zu runden, damit solche Differenzen gar nicht auftreten. Hierzu gibt es in Excel eine einfache Möglichkeit, die in vielen Fällen weiterhilft:

- ☞ Aktivieren Sie in dem Menu „Extras, Optionen“ auf der Karte „berechnen“ die Option „Genauigkeit wie angezeigt“. Die folgende Warnung, können Sie in der Regel ignorieren (mit OK bestätigen), **wenn Sie für alle Berechnungen Formeln eingetragen haben**. Das Programm rundet jetzt alle Zahlen auf die angezeigte Anzahl von Dezimalstellen. Vorsicht ist angebracht bei der Formatierung von **Eingabe-Daten**: *entfernte Dezimalstellen sind endgültig weg!*

Die andere Möglichkeit ist etwas aufwendiger, gestattet es aber, bei jeder Formel individuell zu entscheiden, ob und wie gerundet werden soll:

Geben Sie in dem obigen Beispiel in der Zelle C3 die folgende Formel ein (stellen Sie vorher die Option „Genauigkeit wie angezeigt“ wieder aus):

=RUNDEN(B3*\$C\$2;2)

Die Zahl wird auf zwei Dezimalstellen kaufmännisch gerundet, unabhängig davon, wie viele Dezimalstellen auf dem Bildschirm angezeigt werden.

Die zwei nach dem Semikolon steht dabei für die Anzahl der Nachkommastellen. Diese Zahl muss immer angegeben werden. Wollen Sie auf einen glatten DM-Betrag runden, dann können Sie eine Null eintragen.

Bei der Angabe der Nachkommastellen können Sie auch negative Zahlen angeben, dann rundet das Programm auf Stellen vor dem Komma. Geben Sie in der Zelle D7 z. B. folgende Formel ein:

=RUNDEN(SUMME(D3:D6);-1)

Das Ergebnis ist **20,00 DM**, weil das Programm jetzt auf glatte 10 DM rundet.

Verwandte Funktionen, die nach dem gleichen Schema einzugeben sind:

=ABRUNDEN(**rundet** eine Zahl auf die angegebene Anzahl von Dezimalstellen (nach unten) **ab**.

=AUFRUNDEN(**rundet** eine Zahl auf die angegebene Anzahl von Dezimalstellen (nach oben) **auf**.

=KÜRZEN(funktioniert genau wie **abrunden**.

=GERADE(eine Zahl wird auf die nächste ganze **gerade Zahl** aufgerundet (zur Erinnerung: gerade Zahlen sind 2, 4, 6, 8, 10, . . .).

=UNGERADE(ähnlich wie GERADE. es wird dann eben wird auf die nächste ganze **ungerade** Zahl aufgerundet (1, 3, 5, . . .), übrigens unabhängig vom Vorzeichen.

=UNTERGRENZE(ZAHL; SCHRITT) rundet eine Zahl auf das nächste, durch **Schritt teilbare Vielfache** ab. Wieder ein Beispiel zur obigen Tabelle. In Zelle D7 stehe folgende Formel:

=UNTERGRENZE(SUMME(D3:D6);5)

Das Ergebnis ist **15,00 DM**, weil das Programm jetzt auf glatte 5 DM abrundet.

Ganz entsprechend rundet die Funktion

=OBERGRENZE(

auf das durch die **Schritt**weite teilbare Vielfache auf!

3 STATISTISCHE FUNKTIONEN

In dieser Kategorie sind einige sehr wichtige Funktionen zusammengefasst, die in Excel-Tabellen häufig Anwendung finden.

3.1 Größter Wert: MAX

Diese und die folgenden Funktionen basieren auf demselben Eingabeformat wie die Summenfunktion. Benutzen Sie für das folgende Beispiel die bei der Summenfunktion eingegebene Tabelle. Ergänzen Sie in der Zelle A6 den Text „Größter Wert“. Markieren Sie dann die Zelle B6. Die Tabelle sieht dann so aus:

	A	B	C	D
1		Produkt 1	Produkt 2	Summe
2	<i>Jan</i>	100	1000	1100
3	<i>Feb</i>	150	900	1050
4	<i>Mrz</i>	125	950	1075
5	<i>Summe</i>	375	2850	3225
6	<i>Größter Wert</i>			

Abbildung 7

Klicken Sie auf das Symbol für den Funktionsassistenten und wählen Sie die Funktion MAX aus. Klicken Sie dann auf **Weiter**. Es erscheint folgendes Dialogfenster:

Abbildung 8

Klicken Sie nun in das weiße Feld nach **Zahl1**. Sie erhalten nun eine kurze Erklärung, was Sie in dieses Feld eintragen müssen.

Sie könnten nun die Zelladressen, die addiert werden sollen, hier eintragen. Es sind dies die Zellen B2 bis B5, also **B2:B5**.

Einfacher ist folgende Methode:

Schieben Sie mit der Maus das Dialogfenster zur Seite (klicken Sie dazu in die Titelleiste des Dialogfensters und halten die linke Maustaste gedrückt). Die Zellen, die Sie markieren wollen, müssen dann sichtbar sein (Das Dialogfenster kann ruhig teilweise aus dem Bild verschwinden).

Markieren Sie nun in der Tabelle die Zellen B2 bis B5. Die Zellen werden mit einer durchbrochenen, blinkenden Linie umrahmt. Gleichzeitig erscheinen in dem weißen Bereich des Dialogfensters die Adressen der markierten Zellen.

Mit dieser Methode sollten Sie auch folgende Aufgabe lösen können. Geben Sie diese Tabelle ein:

	A	B
1	Produktionszahlen	
2		
3	Januar	1.000
4	Februar	1.500
5	März	1.600
6	1. Quartal	4.100
7	April	2.000
8	Mai	1.200
9	Juni	1.300
10	2. Quartal	4.500
11	höchster Wert im 1. Halbjahr	

Abbildung 9

Erste Möglichkeit:

- ☞ Rufen Sie den Funktionsassistenten auf und wählen Sie die Funktion MAX aus.
- ☞ Markieren Sie anschließend die Zellen B3 bis B5.
- ☞ Klicken Sie in das Feld neben „Zahl2“ (am einfachsten mit der <TAB>-Taste) und markieren Sie die Zellen B7 bis B9.

zweite Möglichkeit:

- ☞ Tippen Sie ein **=max(**
- ☞ Markieren Sie die Zellen B3 bis B5
- ☞ Drücken Sie die Taste <Strg> und halten Sie diese gedrückt.
- ☞ Markieren Sie die Zellen B7 bis B9 (jetzt <Strg> loslassen).
- ☞ Geben Sie ein: Klammer zu und <Enter>

Die Taste <Strg> funktioniert auch bei Benutzung des Funktionsassistenten. Sie können auf diese Art auch mehrere nicht zusammenhängende Zellbereiche markieren.

3.2 Verwandte Funktionen, die nach demselben Prinzip einzugeben sind

- =MIN(** berechnet den kleinsten Wert der markierten Zahlen
- =MITTELWERT(** berechnet den Mittelwert (Durchschnitt) der markierten Zahlen
- =ANZAHL(** berechnet die Anzahl der Zahlen im markierten Bereich
- =ANZAHL2(** berechnet die Anzahl der Werte (alle nicht leeren Zellen) im markierten Bereich.
- =ANZAHLLEEREZELLEN(** gibt die Anzahl der leeren Zellen im markierten Bereich an.

3.3 Die Funktionen KGRÖSSTE und KKLEINSTE

Ergänzen Sie die Tabelle aus dem vorherigen Kapitel um zwei Zeilen:

	A	B
1	Produktionszahlen	
2		
3	Januar	1.000
4	Februar	1.500
5	März	1.600
6	1. Quartal	4.100
7	April	2.000
8	Mai	1.200
9	Juni	1.300
10	2. Quartal	4.500
11	höchster Wert im 1. Halbjahr	2.000
12	zweitgrößter Wert im 1. Halbjahr	
13	zweitgrößter Wert im 1. Halbjahr	

Abbildung 10

Markieren Sie die Zelle B12 und rufen Sie den Funktionsassistenten auf. Suchen Sie die Funktion KGRÖSSTE aus. Sie erhalten folgendes Dialogfenster:

KGRÖSSTE

Matrix = Zahl

K = Zahl

=

Liefert den k-größten Wert einer Datengruppe.

Matrix ist die Matrix oder der Datenbereich, deren k-größten Wert Sie bestimmen möchten.

Formelergbnis =

Ende Abbrechen

Abbildung 11

Der Cursor sollte neben dem Feld Matrix blinken. Gehen Sie bei der Eingabe folgendermaßen vor:

- ☞ Tippen Sie zunächst eine Klammer auf ein!
- ☞ Markieren Sie die Zellen B3 bis B5.
- ☞ Halten Sie die <Strg>-Taste gedrückt.
- ☞ Markieren Sie B7 bis B9, Klammer zu!
- ☞ Gehen Sie in das Feld neben K und tippen Sie dort eine zwei ein, dann <Enter>.

In der Bearbeitungsleiste sieht die Formel so aus:

=KGRÖSSTE((B3:B5;B7:B9);2)

Die Klammern um die markierten Zellbereiche sind notwendig, damit das Programm erkennt, dass alle Zellen zur Matrix gehören sollen. Ohne die Klammern würde Excel nach dem ersten Semikolon eine Zahl (in unserem Falle die zwei) erwarten und anderenfalls eine Fehlermeldung ausgeben. Wenn Sie einen zusammenhängenden Zellbereich markieren, können Sie die Klammern weglassen.

Wollen Sie den *drittgrößten* Wert berechnen geben Sie anstelle der zwei eine *drei* ein, für den *viertgrößten* eine *vier* usw.

Berechnen Sie anschließend in der Zelle B13 den zweitkleinsten Wert. Die entsprechende Funktion dazu ist:

=KKLEINSTE(

Sie funktioniert genauso wie die Funktion KGRÖSSTE.

3.4 Die Funktion RANG

Ergänzen Sie die Tabelle aus dem letzten Kapitel um eine weitere Spalte:

	A	B	C
1	Produktionszahlen		
2			Rangfolge
3	Januar	1.000	
4	Februar	1.500	
5	März	1.600	
6	1. Quartal	4.100	
7	April	2.000	
8	Mai	1.200	
9	Juni	1.300	
10	2. Quartal	4.500	
11	höchster Wert im 1. Halbjahr	2.000	
12	zweitgrößter Wert im 1. Halbjahr	1.600	
13	zweitgrößter Wert im 1. Halbjahr	1.200	

Abbildung 12

In dieser Spalte soll die Rangfolge der Zahlen in der Spalte B ermittelt werden. Die größte Zahl erhält eine 1, die zweitgrößte eine 2 usw. Dazu markieren Sie zunächst die Zelle C3 und suchen Sie die Funktion Rang mit Hilfe des Funktionsassistenten und geben Sie die folgenden Werte ein:

Zahl: Es muss die Zelle mit dem Wert markiert werden, dessen Rang bestimmt werden soll. In der Zelle C3 soll der Rang der Produktionszahlen vom Januar stehen, also markieren Sie B3

Bezug: Hier werden alle Zellen markiert, deren Rang bestimmt werden soll. Bei nicht zusammenhängenden Zellbereichen müssen Sie Klammern verwenden. Außer-

dem darf sich dieser Bereich nachher beim Auto-Ausfüllen nicht verändern. Also mit F4 festsetzen!

Reihenfolge: Wenn Sie hier eine 1 eingeben, wird der Rang vom kleinsten zum größten Wert vergeben. Die kleinste Zahl bekäme eine 1, die zweitkleinste eine 2 usw. In der Regel lassen Sie dieses Feld also leer.

Abbildung 13

Stören Sie sich nicht an der Fehlermeldung #Wert!. Die Formel funktioniert so. Als Ergebnis sollte eine 6 als Ergebnis erscheinen. Das bedeutet, die Produktionszahlen vom Januar waren der sechstgrößte Wert im 1. Halbjahr.

4 FUNKTIONEN ZU DATUM UND ZEIT

Oft besteht in Excel das Bedürfnis, Zeiträume zwischen zwei Daten zu berechnen. In Excel geht dies durch einfache Subtraktion. Z. B.:

	A	B	C
1	Anfangsdatum	01.01.90	
2	Enddatum	31.12.95	
3	Zeitraum dazwischen:	2190	Tage

Abbildung 14

In Zelle B3 steht die Formel: Enddatum – Anfangsdatum, also

= B2 – B1

☞ Wenn Sie diese Formel eingeben, erscheint vermutlich zunächst ein **Datum**, da Excel automatisch die Formatierung aus den darüber liegenden Zellen übernimmt. Stellen Sie das Format, z. B. über **Format, Zellen, Zahlen, Zahl** um und es müsste die Zahl erscheinen. Das Ergebnis ist die Anzahl der Tage zwischen den beiden Daten.

4.1 Die Funktion T A G E 3 6 0

Um Zinsberechnungen durchzuführen, hätte man aber gerne die Möglichkeit, diese Tage nach dem banküblichen Verfahren (1 Monat= 30 Tage, 1 Jahr = 360 Tage) zu berechnen. Hierfür gibt es in Excel die Funktion Tage360. Ändern Sie in dem Beispiel die Formel in Zelle B3 um in:

=TAGE360(B1;B2;1)

das Ergebnis ist 2159.

In der Funktion wird zuerst das Anfangsdatum, dann das Enddatum angegeben. Die 1 müssen Sie ergänzen, wenn die Berechnung nach dem in Deutschland üblichen Verfahren durchgeführt werden soll (aus dem 31. eines Monats wird der 30., würden Sie eine Null schreiben, so wird der 31. eines Monats in den 1. des Folgemonats umgewandelt.)

Zu beachten ist auch, dass für den ersten und letzten Tage des Zeitraumes zusammen nur ein Tag berechnet wird (Das ist so auch mathematisch korrekt, aber manche Unternehmen zählen in ihren Abrechnungen gern beide Tage als volle Tage, weil es mehr Geld einbringt).

Zu beachten ist, dass die errechneten Zahlen u. U. zunächst als Daten formatiert erscheinen. Wandeln Sie daher die Formate mit Hilfe der Befehle „*Format, Zellen*“, Registerkarte „*Zahler*“ in das „Standard“-Zahlenformat um.

4.2 Weitere Funktionen zu Datum und ZEIT

Weitere verwandte Funktionen erläutert die folgende kleine Tabelle. Geben Sie diese auf einem Tabellenblatt ein und benutzen Sie die angegebenen Funktionen.

	A	B	C	D	E	F	G
1							
2	Heute ist der	2. Juli 1998			mit Uhrzeit:	02.07.98 07:59	
3							
4	Es ist der	2	te Tag		Es ist genau	7	Uhr
5	des	7	ten Monates			59	Minuten
6	im	1998	sten Jahr.			16	Sekunden
7							
8	Das Datum ist:	02.07.98			Die Zeit ist:	7:59:16	
9							

Abbildung 15

Die zugehörigen Funktionen entnehmen Sie der folgenden Abbildung:

	A	B	C	D	E	F	G
1							
2	Heute ist der	=HEUTE()			mit Uhrzeit:	=JETZT()	
3							
4	Es ist der	=TAG(B2)	te Tag		Es ist genau	=STUNDE(F2)	Uhr
5	des	=MONAT(B2)	ten Monates			=MINUTE(F2)	Minuten
6	im	=JAHR(B2)	sten Jahr.			=SEKUNDE(F2)	Sekunden
7							
8	Das Datum ist:	=DATUM(B6;B5;B4)			Die Zeit ist:	=ZEIT(F4;F5;F6)	
9							

5 LOGIK-FUNKTIONEN

5.1 Logische Operatoren

Ein logischer Operator vergleicht zwei Zustände miteinander. Das Ergebnis eines solchen Vergleiches ist, dass der Vergleich entweder wahr oder falsch ist.




Abbildung 18

Im Funktionsassistenten finden Sie unter der WENN-Funktion diese drei Bestandteile wieder. Gehen Sie davon aus, dass Sie in der Praxis **immer alle drei Teile** (Prüfung, Dann-Fall, Sonst-Fall) angeben müssen.

Es genügt, die Funktion einmal einzutragen und mit Hilfe von „Auto-Ausfüllen“ in die darunter liegenden Zellen zu kopieren.

Anschließend können Sie die Ergebnisse noch als „Währung“ formatieren.

Das Ergebnis müsste dann so aussehen:

	A	B	C
1	Name	Alter (Jahre)	Gebühr
2	Hans	15	60 DM
3	Fritz	19	100 DM
4	Lisa	20	100 DM
5	Rosi	13	60 DM

Abbildung 19

5.2.2 Umkehrung der Bedingung

Anstelle der genannten Formel hätte man den Vergleich auch anders wählen können, indem man zunächst fragt:

„Ist das Mitglied nicht älter als 18 Jahre?“

bzw.

„Ist das Mitglied jünger oder gleich 18 Jahre alt?“

Beachten Sie dabei, dass für die Umkehrung der logischen Vergleiche die folgenden Regeln gelten:

Vergleich	Symbol	Umkehrung	Symbol
kleiner als	<	größer oder gleich	>=
größer als	>	kleiner oder gleich	<=
gleich	=	ungleich	<>

Abbildung 20

Der Vergleich lautet in unserem Beispiel dann:

Ist das Mitglied **jünger oder gleich 18 Jahre** alt? → Bedingung

Falls Ja, **dann** beträgt die Gebühr 60 DM, → Dann-Fall

sonst beträgt die Gebühr 100 DM



Sonst-Fall

Aus der oben genannten WENN-Funktion würde damit:

= WENN(B2 <= 18; 60; 100)

Das Ergebnis ist dasselbe wie zuvor. Es kann jedoch in manchen Fällen sinnvoll sein, sich zunächst die Umkehrung der Bedingung notieren und zu prüfen, ob die Funktion dadurch möglicherweise vereinfacht wird.

Dies ist besonders wichtig bei verschachtelten WENN-Funktionen

5.2.3 Ein weiteres Beispiel

In einer WENN-Funktion können der Dann- sowie der Sonst-Fall auch aus einer (fast) beliebig langen Formel bestehen. Dazu ein Beispiel:

Der Club hat sich zu einer etwas aufwendigeren Berechnung der Mitgliedsbeiträge entschlossen. Künftig soll gelten:

- ☞ Der Beitrag beträgt grundsätzlich 50 DM. Ist ein Mitglied älter als sechzehn Jahre, so zahlt es für jedes zusätzliche Lebensjahr 50 Pf.

Das bedeutet anders ausgedrückt:

- ☞ Mitglieder bis 16 Jahre zahlen 50 DM,
- ☞ Mitglieder über 16 Jahre zahlen 50 DM plus 50 Pf für jedes Lebensjahr über 20.

In der Prüfung muss also wiederum zunächst nach dem Alter gefragt werden:

= WENN(B2 <= 16; 50;

Bis dahin ist es einfach. Der Sonst-Fall muss nun die Berechnung des Mitgliedsbeitrages für die über 16jährigen Mitglieder enthalten. Diese Formel würde - unabhängig von der WENN-Funktion - lauten:

= 50 + (B2 - 16) * 0,5.

Diese Formel muss nun als Sonst-Fall in die WENN-Funktion eingetragen werden. Die fertige Formel lautet somit:

= WENN(B2 <= 16; 50; 50 + (B2 - 16) * 0,5)

Enthalten der Dann-Fall und der Sonst-Fall gleichlautende Bestandteile in der Formel, so können diese unter Umständen aus der Formel herausgezogen werden. Bei der zuletzt genannten Formel könnte man die 50 DM, die ja jedes Mitglied zunächst einmal bezahlen muss, ausklammern. Die umgestellte Formel würde somit lauten:

= 50 + WENN(B2 <= 16; 0; (B2 - 16) * 0,5)

5.2.4 Mehr als zwei möglichen Ergebnissen (Verschachtelte WENN-Funktion)

Verändern wir die Bedingung in unserem obigen Beispiel wie folgt:

- Mitglieder die älter als 18 Jahre sind, zahlen 100 DM
- Mitglieder die jünger als 14 Jahre sind, zahlen 40 DM
- Alle übrigen Mitglieder zahlen 60 DM

Wie haben nun **drei mögliche Fälle** zu berücksichtigen. Schreiben Sie sich diese immer erst auf und machen Sie sich klar, **welche** drei Möglichkeiten es gibt!

Es sind in unserem Beispiel folgende:

1. Das Mitglied ist jünger als 14 Jahre, dann zahlt es **40 DM**
2. Das Mitglied ist zwischen 14 und 18 Jahre alt, dann zahlt es **60 DM**
3. Das Mitglied ist älter als 18 Jahre, dann zahlt es **100 DM**.

Mit Hilfe der WENN-Funktion kann man diesen Fall mit Excel lösen, indem man zwei Vergleiche ineinander **verschachtelt**. Dazu müssen Sie zunächst einen der drei Fälle als Bedingung wählen und anschließend für den Dann-Fall oder den Sonst-Fall eine weitere WENN-Funktion wählen.

Für das Beispiel bedeutet dies in Klartext folgenden Vergleich:

Ist das Mitglied **jünger als 14 Jahre**, → erste Prüfung

Falls Ja, **dann** beträgt der Beitrag 40 DM,

sonst gibt es zwei Möglichkeiten:

ist das Mitglied **älter als 18 Jahre**? → zweite Prüfung

Falls Ja, **dann** beträgt der Beitrag 100 DM

sonst beträgt der Beitrag 60 DM

Dieses zweite „sonst“ berücksichtigt die Möglichkeit, dass das Mitglied **zwischen 14 und 18 Jahren alt** ist.

Damit können wir die WENN-Funktion zunächst bis zum Dann-Fall notieren:

```
= WENN(B2 < 14; 40;
```

Für den **Sonst-Fall** haben wir nun **zwei Möglichkeiten**. Da wir aber nur **einen Wert** in die WENN-Funktion als Sonst-Fall eintragen dürfen, müssen wir dort anstelle eines festen Wertes **eine weitere WENN-Funktion** eintragen.

```
= WENN(B2 < 14; 40; SONST-FALL)
```

In diese Klammer müssen wir nun die zweite Prüfung (s.o.) eintragen. Er lautet:

```
WENN(B2 > 18; 100; 60)
```

(in diesem Fall kein Gleichheitszeichen benutzen).

Die fertige verschachtelte WENN-Funktion lautet somit:

```
= WENN(B2 < 14; 40; WENN(B2 > 18; 100; 60))
```

Tragen Sie diese Formel in die Zelle C2 Ihrer Tabelle ein und füllen Sie die Spalte C mit Hilfe von „Auto-Ausfüllen“ aus.

Das Ergebnis sollte so aussehen:

	A	B	C
1	Name	Alter (Jahre)	Gebühr
2	Hans	15	60 DM
3	Fritz	19	100 DM
4	Lisa	20	100 DM
5	Rosi	13	40 DM

Abbildung 21

Die verschachtelte WENN-Funktion ist erscheint auf den ersten Blick kompliziert. Aus diesem Grunde sollten Sie schon bei der Notierung der drei möglichen Vergleiche gewisse Regeln einhalten:

1. Notieren Sie die Fälle in einer logischen Reihenfolge (also z. B., wie in unserem Beispiel, nach dem Alter sortiert).
2. Wählen Sie anschließend den in Ihrer Liste zuerst aufgeführten Vergleich aus.
3. Wählen Sie die Prüfung so, dass der Dann-Fall eindeutig ist (Probieren Sie eventuell die Umkehrung des logischen Vergleiches).

Prinzipiell können Sie auch mehr als drei Möglichkeiten mit Hilfe der WENN-Funktion berücksichtigen. Sie müssen die Funktion dann weiter verschachteln, also z. B. den „sonst“-Fall der inneren WENN-Funktion durch eine weitere WENN-Funktion ersetzen usw.

5.2.5 Die verschachtelte WENN-Funktion mit dem Funktionsassistenten

Nutzen wir diese Beschreibung zu einem weiteren Beispiel:

Die Tabelle soll ergänzt werden um Spalte „Bemerkung“.

	A	B	C	D
1	Name	Alter (Jahre)	Gebühr	Bemerkung
2	Hans	15	60 DM	
3	Fritz	19	100 DM	
4	Lisa	20	100 DM	
5	Rosi	13	40 DM	

Abbildung 22

Dort soll stehen:

„*Junger Hüpfen*“ bei Mitgliedern unter 14 Jahren und
 „*Schon erwachsen?*“ bei Mitgliedern über 18 Jahren.

Dies soll mit Hilfe des Funktionsassistenten eingegeben werden. Markieren Sie also die erste Zelle unter der Überschrift „Bemerkung“ (D2) und klicken Sie auf das Symbol „Funktionsassistent“. Suchen Sie die WENN-Funktion aus und geben Sie Prüfung und dann-Fall ein:

WENN

Prüfung: B2<14 = FALSCH

Dann_Wert: "Junger Hüpfen" = "Junger Hüpfen"

Sonst_Wert: = Beliebig

Gibt eine Wahrheitsprüfung an, die durchgeführt werden soll.

Sonst_Wert ist das Resultat der Funktion, wenn die Wahrheitsprüfung FALSCH ergibt.

Formelergbnis =FALSCH

Ende Abbrechen

Abbildung 23

Haben Sie daran gedacht, dass **Texte innerhalb von Funktionen grundsätzlich in Anführungszeichen** zu setzen sind? Wenn nicht, dann macht das Programm dies selber, sobald Sie in das Feld „Sonst-Wert“ klicken.

Wenn der Cursor in dem dritten Feld blinkt, müssten Sie nun die zweite WENN-Funktion eingeben. Gehen Sie dabei folgendermaßen vor:

Links neben der Bearbeitungsleiste erscheint anstelle der sonst üblichen Zelladresse ein Drop-Down-Menü, in dem Sie die zuletzt benutzten Funktionen auswählen können. Das „wenn“ erscheint als erstes in der Liste. Sie brauchen also nur auf diese Schaltfläche „wenn“ zu klicken, das Funktionsfenster öffnet sich wieder und Sie können die zweite WENN-Funktion komplett eintragen. Beim Sonst-Wert tragen Sie zwei Anführungsstriche ein (oder einmal die Leertaste drücken). Dies bedeutet, dass in dem Falle nichts in der Zelle stehen soll.




Abbildung 24

In der Bearbeitungsleiste erscheint die komplette Formel:

```
=WENN(B2<14;"JUNGER HÜPFER";
WENN(B2>18;"SCHON ERWACHSEN?"; ""))
```

- ☞ Müssen Sie eine der beiden WENN-Funktionen **korrigieren**, so klicken Sie *in der Bearbeitungsleiste auf diese Funktion*. Das Fenster des Funktionsassistenten wechselt entsprechend.

Nachdem Sie die Formel nach unten kopiert haben, sollten die Ergebnisse in der Tabelle erscheinen:

	A	B	C	D
1	Name	Alter (Jahre)	Gebühr	Bemerkung
2	Hans	15	60 DM	
3	Fritz	19	100 DM	<i>Schon erwachsen?</i>
4	Lisa	20	100 DM	<i>Schon erwachsen?</i>
5	Rosi	13	40 DM	<i>Junger Hüpfert</i>

Abbildung 25

5.3 Die Funktion ODER

Die hier und im nächsten Kapitel beschriebenen Funktionen werden in der Regel in Verbindung mit anderen Funktionen, meisten mit der WENN-Funktion, benutzt. Geben Sie zu Erläuterung dieser Funktionen das folgende Beispiel in eine Excel-Tabelle ein:

	A	B	C	D	E	F
1	Rechnung					
2						
3	Artikel	Waren- gruppe	Bestell- menge	Einzel- preis	Gesamt- preis	Rabatt [DM]
4	A01	A	1	25,00 DM	25,00 DM	
5	A02	B	10	10,00 DM	100,00 DM	
6	A03	C	5	2,50 DM	12,50 DM	
7	A04	D	50	5,00 DM	250,00 DM	
8	A05	A	100	2,50 DM	250,00 DM	
9	A06	C	500	1,00 DM	500,00 DM	
10	A07	E	1.000	1,50 DM	1.500,00 DM	
11	A08	G	25	5,00 DM	125,00 DM	
12	A09	E	10	10,00 DM	100,00 DM	
13	A10	F	75	20,00 DM	1.500,00 DM	
14	A11	H	250	7,50 DM	1.875,00 DM	

Abbildung 26

Für die Berechnung des Rabattes sollen folgende Bedingungen gelten:

Es gibt 3% Rabatt auf den Gesamtpreis, sofern

- ☞ der Gesamtpreis mindestens 1000 DM beträgt oder
- ☞ die Bestellmenge wenigstens 500 Stück beträgt oder
- ☞ der Artikel aus der Warengruppe E stammt.

Wenn Sie diese Aufgabe mit Hilfe einer verschachtelten WENN-Funktion lösen wollen, würde diese z. B. so aussehen:

**= WENN(E4 >= 1000; 3%;
WENN(C4 >= 500; 3%;
WENN(B4 = "E"; 3%; 0))) * E4**

Diese Formel lässt sich mit Hilfe der ODER-Funktion vereinfachen zu einer nicht verschachtelten WENN-Funktion. Anstelle einer einzelnen Prüfung wird die ODER-Funktion eingegeben. In den Klammern der ODER-Funktion stehen dann die drei Prüfungen:

**= WENN(ODER(E4 >= 1000; C4 >= 500;
B4 = "E"); 3% * E4; 0)**

In Worten ausgedrückt bedeutet diese Funktion:

Wenn entweder E4 >= 1000 **oder** C4 >= 500 **oder** B4 = "E" ist,
dann berechne 3% von E4 als Rabatt,
sonst berechne keinen Rabatt.

Auch diese WENN-Funktion besteht aus **Prüfung**, **dann-Fall** und **sonst-Fall**. Die Prüfung ist dabei die ODER-Funktion.

Kompletieren Sie die Tabelle mit Hilfe von Auto-Ausfüllen und kontrollieren Sie die Ergebnisse. Es wird immer dann ein Rabatt berechnet, wenn **wenigstens eine** der drei Bedingungen erfüllt ist.

5.4 Die Funktion UND

Ändern Sie die Formel in der Tabelle aus dem Kapitel „Die Oder-Funktion“ ab, indem Sie das **ODER** ersetzen durch **UND**:

= WENN(UND(E4 >= 1000; C4 >= 500; B4 = "E"); 3% * E4; 0)

Füllen Sie die Spalte mit dieser Formel aus und schauen Sie sich die Ergebnisse an. Es wird nur noch bei Artikel A07 ein Rabatt gewährt.

Grund: Nur bei diesem Artikel sind **alle drei Bedingungen gleichzeitig** erfüllt.

Während bei der *ODER-Funktion eine von den drei Bedingungen erfüllt sein* muss, damit der dann-Fall zum tragen kommt, müssen bei der *UND-Funktion* dafür *alle drei Bedingungen gleichzeitig* erfüllt sein.

5.5 Die Funktion NICHT

Es gibt außerdem noch die Funktion NICHT, die sich aber meist durch Umkehrung einer Prüfung in der WENN-Funktion umgehen lässt.

= WENN(NICHT(E4 < 1000); 3% * E4; 0)

bedeutet z. B., dass 3% Rabatt berechnet werden, wenn der Gesamtpreis nicht unter 1000 DM liegt.

5.6 Die Funktionen ZÄHLENWENN und SUMMEWENN

ergänzen Sie die Tabelle aus Abbildung 26 auf Seite 21 um die folgende kleine Tabelle:

17	Bestellung nach Warengruppen		
18			
19	Warengruppe	Anzahl	Summe
20	A		
21	B		
22	C		
23	D		
24	E		
25	F		
26	G		
27	H		
28			

Abbildung 27

In der markierten Zelle geben Sie nun folgende Formel ein:

=ZÄHLENWENN (\$B\$4:\$B\$14;A20)

Das Ergebnis ist 2. Diese Funktion zählt, wie oft der Inhalt der Zelle A20 („A“) im Bereich B4 Bis B14 (das sind die Warengruppen in der darüber stehenden Tabelle) vorkommt. Warengruppe A ist genau zweimal vertreten.

Ergänzen Sie in der Zelle daneben (C20) die folgende Funktion SUMMEWENN. Füllen Sie die Felder im Funktionsassistenten folgendermaßen aus:

SUMMEWENN

Bereich	\$B\$4:\$B\$14	= {"A";"B";"C";"D";"A";
Suchkriterien	A20	= "A"
Summe_Bereich	\$E\$4:\$E\$14	= {25;100;12,5;250;2!

= 275

Addiert Zahlen, die mit den Suchkriterien übereinstimmen.

Summe_Bereich sind die Zellen, die Sie summieren wollen.

Formelergebnis = 275,00 DM

Ende Abbrechen

Abbildung 28

Bei dieser Funktion passiert folgendes:

- ☞ Die Funktion sucht im Bereich B4 bis B14 (die Warengruppen) das Suchkriterium (Inhalt der Zelle A20, also zunächst das A).
- ☞ Die zugehörigen Beträge (Eingabe im Feld „Summe_Bereich“, also die Beträge in den Zellen E4 bis E14) werden dann addiert. Das A findet das Programm zweimal, an erster und an fünfter Stelle in der Liste B4 bis B14, also addiert es die Beträge, die an erster und fünfter Stelle der Liste E4 bis E14 stehen.

Das Ergebnis beträgt für die Warengruppe A : 25 DM plus 250 DM = 275 DM.

Nachdem Sie beide Formeln eingetragen haben, markieren Sie die Zellen C20 und D20 und ziehen Sie die Markierung mit Hilfe von Auto-Ausfüllen nach unten. Dies vervollständigt die Tabelle:

	A	B	C
17	Bestellung nach Warengruppen		
18			
19	Warengruppe	Anzahl	Summe
20	A	2	275,00 DM
21	B	1	100,00 DM
22	C	2	512,50 DM
23	D	1	250,00 DM
24	E	2	1.600,00 DM
25	F	1	1.500,00 DM
26	G	1	125,00 DM
27	H	1	1.875,00 DM

Abbildung 29

6 MATRIX-FUNKTIONEN

6.1 Die Funktion SVERWEIS

Stellen Sie sich folgende Aufgabe vor:

Sie haben als Lehrer einer Schulklasse eine Klassenarbeit korrigiert und wollen die Bewertung mit Hilfe von Excel vornehmen. Die Namen der Schüler und die erreichte Punktzahl tragen Sie dazu in eine Excel-Tabelle ein:

	A	B	C
1	Klassenarbeit - Ergebnisse		
2		erzielte Punkte	Note
3	Hans	81	
4	Susi	35	
5	Sandra	78	
6	Theo	90	
7	Udo	32	
8	Peter	75	
9	Lisa	85	
10	Uli	62	
11	Erika	57	
12	Rebecca	87	
13	Ingrid	71	
14	Klaus	45	
15			
16	maximale Punktzahl	90	

Abbildung 30

In die markierte Zelle soll nun eine Formel eingetragen werden, mit deren Hilfe die Note automatisch ermittelt wird. Dazu gelte folgende Bewertungstabelle:

	<i>erreicht Punktzahl</i>	<i>Note</i>
ab	0	<i>ungenügend</i>
ab	20	<i>mangelhaft</i>
ab	45	<i>ausreichend</i>
ab	60	<i>befriedigend</i>
ab	75	<i>gut</i>
ab	85	<i>sehr gut</i>

Abbildung 31

Mit den Ihnen bislang bekannten Funktionen wäre diese Aufgabe etwa so zu lösen:

```
= WENN(B3 >= 85; "SEHR GUT";
WENN(B3 >= 75; "GUT";
WENN(B3 >= 60; "BEFRIEDIGEND";
WENN(B3 >= 45; "AUSREICHEND";
WENN(B3 >= 20; "MANGELHAFT"; "UNGENÜGEND"))
)))
```

Die Formel funktioniert so! Durch „Auto-Ausfüllen“ kann sie in die darunter liegenden Zellen kopiert werden. Die Eingabe erfordert halt etwas Tipparbeit; außerdem muss man darauf achten, die einzelnen Prüfungen in der korrekten Reihenfolge einzubauen.

Es gibt in Excel allerdings Funktionen, die dieses Problem einfacher lösen und außerdem auch längere Listen als die obigen Notentabelle verwerten können. Geben Sie dazu zunächst die Notentabelle so ein, wie sie oben abgebildet ist. Am besten auf Tabellenblatt 2. Auf Tabellenblatt 1 steht die Tabelle mit den Klausurergebnissen.

Benennen Sie die beiden Tabellenblätter, z. B. „Klausur“ und „Noten“.

markieren Sie die Zelle C3 in der Tabelle „Klausur“ und rufen Sie den Funktionsassistenten auf. Suchen Sie die Funktion **SVERWEIS** heraus. Sie erhalten folgendes Dialogfenster:

SVERWEIS

Suchkriterium = Beliebig

Matrix = Zahl

Spaltenindex = Zahl

Bereich_Verweis = Wahrheitswert

=

Durchsucht die erste Spalte einer Matrix und durchläuft die Zeile nach rechts, um den Wert einer Zelle zurückzugeben.
Suchkriterium ist der Wert, nach dem Sie in der ersten Spalte der Matrix suchen.

 Formelergbnis =

Abbildung 32

In die vier weißen Felder werden folgende Daten eingetragen:

Suchkriterium: In der Regel eine Zelladresse. Diesen Wert soll das Programm nachher in der ersten Spalte der *Matrix* suchen und das zugehörige Ergebnis ausgeben.

In dem Beispiel muss hier die Adresse **B3** stehen. Markieren Sie diese Zelle mit der Maus.

Matrix: Hier wird ein mindestens zwei Spalten breiter Bereich von Zelladressen eingegeben. In der ersten Spalte der Matrix wird nachher der Wert gesucht, den Sie als Suchkriterium markiert haben. In einer weiteren Spalte findet das Programm dann das zugehörige Ergebnis. Dazwischen liegende Spalten sind häufig ohne Belang, **müssen** aber mit markiert werden!

Blättern Sie nun mit der Maus die Tabelle „Noten“ auf und markieren Sie dort die komplette Notentabelle, ohne die Überschrift. Drücken Sie anschließend die Taste F4. In dem Feld müsste dann der folgende Eintrag stehen:

Noten!\$B2:\$C\$7

Wichtig: **Blättern Sie anschließend nicht auf ein anderes Tabellenblatt um!**

Spaltenindex: Das Programm muss nun noch wissen, in der wievielten Spalte der Matrix die möglichen Ergebnisse dieser Berechnung stehen. Auch wenn dies in unserem Beispiel eigentlich klar ist, muss hier die Zahl **2** (für die zweite Spalte in dem markierten Bereich der Notentabelle) stehen.

Bereich-Verweis: In unserem Beispiel können Sie das Feld leer lassen oder eine 1 eintragen. Näheres dazu siehe unten.

Wenn Sie die Einträge vollständig eingegeben haben, müsste in der Bearbeitungsleiste folgende Formel stehen:

=SVERWEIS(B3;NOTEN!\$B\$2:\$C\$7;2)

6.1.1 Einige Regeln zur SVERWEIS-Funktion

Das Programm arbeitet die SVERWEIS-Funktion folgendermaßen ab:

6.1.1.1 erste Möglichkeit: Bereich-Verweis = 1 (oder leer)

Der Wert in der Zelle, die als **Suchkriterium** angegeben wurde, wird in der **ersten Spalte der Matrix** gesucht. Dabei durchläuft das Programm die Matrix *von oben nach unten*. Findet das Programm einen Wert, der **größer ist als das Suchkriterium, dann geht es eine Zeile nach oben** und **nach rechts zu der beim Spaltenindex angegebenen Spalte**. Den Wert in der so erreichten Zelle gibt das Programm als Ergebnis aus.

Wichtig: Die **erste Spalte** der Matrix **muss** in diesem Falle aufsteigend (vom kleinsten zum größten Wert) **sortiert** sein (Numerisch oder alphabetisch)!

6.1.1.2 zweite Möglichkeit: Bereich-Verweis = 0

Wird als Bereich-Verweis eine Null eingegeben, sucht das Programm **in der ersten Spalte der Matrix nach dem exakten Wert**, der als Suchkriterium angegeben wurde. Sobald das Programm den Wert findet, wird das Ergebnis ausgegeben.

Findet das Programm den Wert nicht, so gibt es die Fehlermeldung #NV (nicht vorhanden!) aus. Taucht der Wert **mehrfach** in der ersten Spalte der Matrix auf, so wird das Ergebnis ausgegeben, das dem **ersten Eintrag** in der Matrix zugeordnet ist.

6.2 Die Funktion WVERWEIS

Geben Sie folgende Tabelle ein:

	A	B	C
1	Name	Gehaltsstufe	Gehalt
2	Amann	B	
3	Bemann	C	
4	Cemann	A	
5	Demann	C	
6	Emann	E	

Abbildung 33

Das Gehalt soll mit Hilfe einer Verweis-Funktion ermittelt werden. Auf einer zweiten Tabelle geben Sie dazu die folgende Gehaltstabelle ein:

	A	B	C	D	E	F	G	H
1	Gehaltsstufe	A	B	C	D	E	F	G
2	Grundgehalt	1.600 €	1.800 €	2.000 €	2.350 €	2.700 €	3.200 €	3.700 €

Abbildung 34

Nennen Sie das erste Tabellenblatt „Mitarbeiter“, das zweite „Gehaltsstufen“.

In die Zelle C2 der Mitarbeiter-Tabelle muss nun eine Formel eingetragen werden, die automatisch die Grundgehälter entsprechend der Gehaltsstufen zuordnet.

Mit SVERWEIS funktioniert dies deshalb nicht, weil diese Funktion eine Matrix erfordert, bei der die Liste der Suchkriterien (Gehaltsstufen) und der möglichen Ergebnisse (Grundgehälter) **senkrecht** angeordnet sein müssen (Daher **SVERWEIS**)

Ist die Matrix wie in diesem Beispiel **waagerecht** angeordnet, dann benötigen Sie die Funktion **WVERWEIS**. Anstelle eines Spaltenindex geben Sie einen Zeilenindex (in der wievielten Zeile der Matrix stehen die möglichen Ergebnisse - in unserem Falle die Gehaltsstufen?) an. Vollständig ausgefüllt sollte die Funktion so aussehen:

WVERWEIS

Suchkriterium B2 = "B"

Matrix Gehaltsstuf!\$A\$1:\$H\$2 = {"Gehaltsstufe"."A".}

Zeilenindex 2 = 2

Bereich_Verweis = Wahrheitswert

= 1800

Durchsucht die erste Zeile einer Matrix und durchläuft die Spalte nach unten, um den Wert einer Zelle zurückzugeben.

Bereich_Verweis gibt an, ob eine genaue Übereinstimmung gefunden werden soll.

Formelergebnis = 1800

OK Abbrechen

Abbildung 35

6.3 Die Funktion VERWEIS

In den allermeisten Fällen lassen sich mit Hilfe der Funktionen SVERWEIS bzw. WVERWEIS die oben beschriebenen Probleme lösen. Diese Funktionen sind quasi vereinfachte Nachfolge-Funktionen der „Ur“-VERWEIS-Funktion. Diese ist - in zwei Varianten - auch in der neuesten Version von Excel noch enthalten.

Ein Beispiel, bei dem die Funktionen SVERWEIS oder WVERWEIS *nicht* genutzt werden können, sei hier angeführt:

Geben Sie zunächst folgende Tabelle ein, nennen Sie das Tabellenblatt „Rechnung“:

	A	B	C	D	E	F
1	<i>Rechnung</i>					
2						
3	Artikel-Nr.	Menge	Einzel- preis	Zwischen- summe	Rabatt [DM]	End- betrag
4	xy01	100	1,50 DM	150,00 DM		
5	xy02	1.000	0,95 DM	950,00 DM		
6	xy03	2.000	1,00 DM	2.000,00 DM		
7	xy04	50	2,95 DM	147,50 DM		
8	xy05	25	25,00 DM	625,00 DM		
9	xy06	600	1,95 DM	1.170,00 DM		

Abbildung 36

Die Zwischensumme haben Sie natürlich mit einer einfachen Multiplikation berechnet. Für den Rabatt in DM benötigen Sie weitere Angaben. Die folgende Tabelle mit der Rabattstaffelung sollte auf einem weiteren Tabellenblatt mit dem Namen „Rabatt“ eingegeben werden:

	A	B	C	D	E	F	G
1	Rabattstaffelung:						
2							
3	Rabattsätze		0,0%	1,0%	1,5%	2,0%	2,5%
4							
5							
6	ab	- DM					
7	ab	250 DM					
8	ab	500 DM					
9	ab	1.000 DM					
10	ab	2.000 DM					

Abbildung 37

Markieren Sie die Zelle E4 auf der Tabelle „Rechnung“ und rufen Sie mit dem Funktionsassistenten die VERWEIS-Funktion auf. Sie erhalten zunächst eine Meldung, dass Sie eine von zwei Varianten dieser Funktion wählen sollen. Bestätigen Sie dies einfach mit OK, Sie wählen so die Variante „Suchkriterium, Suchvektor, Ergebnisvektor“ aus.

Im folgenden Dialogfenster müssen Sie folgende Zellbezüge eingeben:

Suchkriterium wie bei SVERWEIS und WVERWEIS; also hier die Zwischensummen – Zelle D4.

Suchvektor Der Zellbereich (eine Zeile oder Spalte) in dem die Suchkriterien aufgelistet sind. In unserem Beispiel sind dies die Preise in der Rabatt-Tabelle.

Ergebnisvektor Der Zellbereich (eine Zeile oder Spalte), in dem die möglichen Ergebnisse stehen. In unserem Beispiel die Rabattsätze.

Suchvektor und Ergebnisvektor müssen gleich lang sein!

Komplett ausgefüllt sieht das Dialogfenster dann so aus:

VERWEIS

Suchkriterium D4 = 150

Suchvektor Rabatt!\$B\$6:\$B\$10 = {0;250;500;1000;20}

Ergebnisvektor Rabatt!\$C\$3:\$G\$3 = {0,0,01,0,015,0,02,

= 0

Durchsucht die Werte eines Vektors oder einer Matrix.

Ergebnisvektor ist ein Bereich, der nur eine Zeile oder Spalte enthält.

Formelergebnis = 0,0%

Ende Abbrechen

Abbildung 38

Durch „Auto-Ausfüllen“ erhalten Sie die kompletten Ergebnisse. Formatieren Sie die Zellen anschließend im Prozentformat mit einer Dezimalstelle.

6.4 Die Funktion VERGLEICH

Eine weitere Funktion aus der Gruppe der Matrixfunktionen ist der VERGLEICH. Dazu ein Beispiel. Betrachten Sie die folgende Tabelle, hier soll ein Preisvergleich für verschiedene Waren durchgeführt werden. Dazu wurden in einer Tabelle die Angebote von vier verschiedenen Händlern zusammengestellt.

	A	B	C	D	E	F	G
1	Preisvergleich						
2							
3		Preise bei Anbieter Nr.				günstigster	am günstigsten
4	Artikel	1	2	3	4	Preis	ist Anbieter:
5	Artikel 1	830,15 €	767,42 €	764,75 €	806,33 €	764,75 €	
6	Artikel 2	311,20 €	318,36 €	284,40 €	321,94 €	284,40 €	
7	Artikel 3	744,48 €	779,27 €	815,31 €	822,63 €	744,48 €	
8	Artikel 4	855,55 €	778,75 €	836,10 €	795,88 €	778,75 €	
9	Artikel 5	570,97 €	596,14 €	536,56 €	603,59 €	536,56 €	
10	Artikel 6	419,95 €	362,10 €	404,39 €	439,14 €	362,10 €	
11	Artikel 7	208,63 €	226,43 €	258,43 €	264,35 €	208,63 €	
12	Artikel 8	667,59 €	685,40 €	720,45 €	632,79 €	632,79 €	
13	Artikel 9	1.079,64 €	1.075,51 €	1.043,51 €	990,92 €	990,92 €	
14							

Abbildung 39

Den günstigsten Preis können Sie rasch mit der Funktion Min ermitteln. Um auch anzuzeigen, welcher Anbieter das jeweils günstigste Angebot hat, reicht es in diesem Falle die Nummer des Anbieters daneben zu schreiben. Die schaffen wir mit Hilfe der Funktion VERGLEICH in Zelle G5:

VERGLEICH

Suchkriterium F5 = 764,7530895

Suchmatrix B5:E5 = {830,149799364748}

Vergleichstyp 0 = 0

= 3

Sucht Werte innerhalb eines Bezuges oder einer Matrix.

Vergleichstyp ist die Zahl 1, 0 oder -1, die angibt, welcher Wert zurückgegeben werden soll. Weitere Informationen über Vergleichstypen finden Sie in der Hilfe.

Formelerggebnis = 3

OK Abbrechen

Abbildung 40

Es wird der Wert, der in Zelle F5 steht innerhalb der Liste B5 bis E5 gesucht. Die Position, an der dieser Wert in der Liste steht, wird als Ergebnis ausgegeben. In diesem Falle steht der Wert an dritter Stelle, daher ist das Ergebnis 3. Das klappt so natürlich nur, wenn die Positionsnummer mit der als Ergebnis gewünschten Anbieter-Nr. identisch ist.

Die 0 beim Vergleichstyp muss angegeben werden, das die Liste B5:E5 nicht aufsteigend sortiert ist! Das ist ähnlich wie beim *Bereichsverweis* bei den VERWEIS-Funktionen.

Geben Sie die Funktion so ein und kopieren Sie diese mit Hilfe von „AutoAusfüllen“ nach unten. Das Ergebnis sollte so aussehen:

	A	B	C	D	E	F	G
1	Preisvergleich						
2							
3		Preise bei Anbieter Nr.				günstigster	am günstigsten
4	Artikel	1	2	3	4	Preis	ist Anbieter:
5	Artikel 1	830,15 €	767,42 €	764,75 €	806,33 €	764,75 €	3
6	Artikel 2	311,20 €	318,36 €	284,40 €	321,94 €	284,40 €	3
7	Artikel 3	744,48 €	779,27 €	815,31 €	822,63 €	744,48 €	1
8	Artikel 4	855,55 €	778,75 €	836,10 €	795,88 €	778,75 €	2
9	Artikel 5	570,97 €	596,14 €	536,56 €	603,59 €	536,56 €	3
10	Artikel 6	419,95 €	362,10 €	404,39 €	439,14 €	362,10 €	2
11	Artikel 7	208,63 €	226,43 €	258,43 €	264,35 €	208,63 €	1
12	Artikel 8	667,59 €	685,40 €	720,45 €	632,79 €	632,79 €	4
13	Artikel 9	1.079,64 €	1.075,51 €	1.043,51 €	990,92 €	990,92 €	4

Abbildung 41

6.5 Die Funktion INDEX

Erweitern Sie die Tabelle aus dem letzten Kapitel noch ein wenig:

	A	B	C	D	E	F	G	H
1	Preisvergleich							
2								
3		Preise bei Anbieter Nr.				günstigster	am günstigsten	
4	Artikel	1	2	3	4	Preis	ist Anbieter:	
5	Name des Händlers:	Edoka	Sparr	Minus	Brutto		Nr.:	Name
6	Artikel 1	830,15 €	767,42 €	764,75 €	806,33 €	764,75 €	3	
7	Artikel 2	311,20 €	318,36 €	284,40 €	321,94 €	284,40 €	3	
8	Artikel 3	744,48 €	779,27 €	815,31 €	822,63 €	744,48 €	1	
9	Artikel 4	855,55 €	778,75 €	836,10 €	795,88 €	778,75 €	2	
10	Artikel 5	570,97 €	596,14 €	536,56 €	603,59 €	536,56 €	3	
11	Artikel 6	419,95 €	362,10 €	404,39 €	439,14 €	362,10 €	2	
12	Artikel 7	208,63 €	226,43 €	258,43 €	264,35 €	208,63 €	1	
13	Artikel 8	667,59 €	685,40 €	720,45 €	632,79 €	632,79 €	4	
14	Artikel 9	1.079,64 €	1.075,51 €	1.043,51 €	990,92 €	990,92 €	4	

Abbildung 42

Jetzt stehen auch die Namen der vier Anbieter in der Tabelle. Diese sollen nun auch in der Spalte H erscheinen. Dies lässt sich mit Hilfe der Funktion INDEX erreichen.

Diese Funktion gibt als Ergebnis den Inhalt einer Zelle wieder, die in einer Liste an einer bestimmten, in der Funktion anzugebenden Position steht. Wenn Sie die Funktion Index aufrufen, haben Sie zunächst die Wahl zwischen zwei Varianten. klicken Sie einfach auf „OK“, Sie wählen dann den Standardtyp dieser Funktion (*Matrix;Zeile;Spalte*). Das entsprechende Dialogfenster sieht dann z. B. so aus:

INDEX

Matrix: = {"Edoka"."Sparr"."Mi
 Zeile: = 1
 Spalte: = 3

= "Minus"

Verwendet einen Index, um aus einem Bezug oder einer Matrix einen Wert zu wählen.

Spalte markiert die Spalte in der Matrix, aus der ein Wert geliefert werden soll.

 Formelerggebnis =Minus

Abbildung 43

folgende Logik verbirgt sich dahinter: In der Matrix B5 bis E5 wird der Inhalt der Zelle in der ersten Zeile (die Matrix ist eine Zeile hoch) und in der „G6-ten“ Spalte gesucht und als Ergebnis ausgegeben. In dem speziellen Fall wird aus der Liste der Anbieter *Edoka-Sparr-Minus-Brutto* der dritte Name ausgegeben.

Wenn Sie die Formel so eingeben und nach unten kopieren, sollte das Ergebnis so aussehen:

	A	B	C	D	E	F	G	H
1	Preisvergleich							
2								
3		Preise bei Anbieter Nr.					günstigster	am günstigsten
4	Artikel	1	2	3	4	Preis	ist Anbieter:	
5	Name des Händlers:	Edoka	Sparr	Minus	Brutto		Nr.:	Name
6	Artikel 1	830,15 €	767,42 €	764,75 €	806,33 €	764,75 €	3	Minus
7	Artikel 2	311,20 €	318,36 €	284,40 €	321,94 €	284,40 €	3	Minus
8	Artikel 3	744,48 €	779,27 €	815,31 €	822,63 €	744,48 €	1	Edoka
9	Artikel 4	855,55 €	778,75 €	836,10 €	795,88 €	778,75 €	2	Sparr
10	Artikel 5	570,97 €	596,14 €	536,56 €	603,59 €	536,56 €	3	Minus
11	Artikel 6	419,95 €	362,10 €	404,39 €	439,14 €	362,10 €	2	Sparr
12	Artikel 7	208,63 €	226,43 €	258,43 €	264,35 €	208,63 €	1	Edoka
13	Artikel 8	667,59 €	685,40 €	720,45 €	632,79 €	632,79 €	4	Brutto
14	Artikel 9	1.079,64 €	1.075,51 €	1.043,51 €	990,92 €	990,92 €	4	Brutto
15								

Abbildung 44

6.5.1 INDEX und VERGLEICH kombiniert

Nun wäre es bei der Tabelle aus Abbildung 42 noch eleganter, wenn man auf die Nummern verzichten würde und nur die Namen der Geschäfte angegeben wären. Das sähe dann im Ergebnis etwa so aus:

	A	B	C	D	E	F	G
1	Preisvergleich						
2							
3	Artikel	Preise bei:				günstigster	Name des Anbieters
4	Name des Händlers:	Edoka	Sparr	Minus	Brutto	Preis	
5	Artikel 1	830,15 €	767,42 €	764,75 €	806,33 €	764,75 €	Minus
6	Artikel 2	311,20 €	318,36 €	284,40 €	321,94 €	284,40 €	Minus
7	Artikel 3	744,48 €	779,27 €	815,31 €	822,63 €	744,48 €	Edoka
8	Artikel 4	855,55 €	778,75 €	836,10 €	795,88 €	778,75 €	Sparr
9	Artikel 5	570,97 €	596,14 €	536,56 €	603,59 €	536,56 €	Minus
10	Artikel 6	419,95 €	362,10 €	404,39 €	439,14 €	362,10 €	Sparr
11	Artikel 7	208,63 €	226,43 €	258,43 €	264,35 €	208,63 €	Edoka
12	Artikel 8	667,59 €	685,40 €	720,45 €	632,79 €	632,79 €	Brutto
13	Artikel 9	1.079,64 €	1.075,51 €	1.043,51 €	990,92 €	990,92 €	Brutto

Abbildung 45

Um dies zu erreichen, müssen die Funktionen INDEX und VERGLEICH verschachtelt werden. Sie ermitteln quasi innerhalb der INDEX-Funktion die Zeilennummer mit Hilfe der VERGLEICH-Funktion:




Abbildung 46

Ausgeschrieben Sie das dann so aus (Inhalt von G5 in Abbildung 45):

=INDEX(\$B\$4:\$E\$4;VERGLEICH(F5;B5:E5;0))

7 FINANZMATHEMATISCHE FUNKTIONEN

In dieser Kategorie befinden sich einige Funktionen, mit deren Hilfe man einige Daten zu Krediten errechnen kann.

7.1 Die Funktion RMZ

Angenommen, sie beabsichtigen, eine größere Anschaffung über einen Kredit zu finanzieren. Sie kennen den Preis, den Zinssatz und die Laufzeit, nach der Sie den Kredit getilgt haben möchten. Die Funktion RMZ (regelmäßige Zahlung) kann Ihnen daraus die Raten berechnen, die auf Sie zukommen.

In der folgenden Tabelle sollen 10.000 € in einem Zeitraum von 5 Jahren (=60 Monaten) getilgt werden. Der Zinssatz betrage 5% (achten Sie auf das Prozentformat!) Die Anzahl der Monate berechnen Sie mit der Formel =B5*12:

	A	B	C
1	Kreditsumme:	10.000 €	
2			
3	Zinssatz:	5%	
4			
5	Laufzeit	5 Jahre	
6		60 Monate	
7			
8	Rate		

Abbildung 47

Die Monatsrate berechnen Sie nun, indem Sie zunächst die Funktion RMZ aufrufen. Geben Sie folgende Daten ein:

RMZ

Zins B3/12 = 0,004166667

Zzr B6 = 60

Bw -B1 = -10000

Zw = Zahl

F = Zahl

= 188,7123364

Liefert die konstante Zahlung einer Annuität pro Periode.

Zins ist der Zinssatz pro Periode (Zahlungszeitraum).

Formelergbnis = 188,7123364

OK Abbrechen

Abbildung 48

- Zins:** ist der Zinssatz pro Zahlungsperiode. Da Sie Ihre Raten monatlich zahlen, der angegebene Zinssatz aber ein Jahreszins ist, müssen Sie B3 durch 12 teilen. Außerdem muss der Zins im Prozentformat angegeben werden!
- Zzr:** bedeutet Zahlungszeitraum. Angegeben werden muss die in derselben Einheit, die für die zu errechnende Rate gelten soll. Da wir Monatsraten als Ergebnis haben wollen, muss als Zeitraum auch 60 (Monate) angegeben werden.
- Bw:** ist der sogenannten Barwert, also die Kreditsumme. Die sollte negativ sein, da es sich um eine Schuld handelt (daher das Minuszeichen davor).

Die restlichen Angaben sind zunächst nicht erforderlich. Sie erhalten als Ergebnis die monatlich zu zahlenden Beträge: **-188,71 €**.

Benutzen wir die Funktion RMZ noch für eine andere Aufgabe. Sie wollen sich einen teuren Wunsch erfüllen, allerdings erst in drei Jahren. Um dann die erforderliche Summe zusammen zu haben, sparen Sie monatlich einen konstanten Betrag. Wie hoch muss dieser Betrag sein.

Geben Sie die Tabelle entsprechend Abbildung 47 ein. Nehmen wir an Sie benötigen 10.000 € in drei Jahren (36 Monaten). Der Zinssatz, den Sie mit Ihren Raten erzielen betrage 1,5%. Geben Sie dann in das Feld, in dem die Monatsraten erscheinen sollen folgende Daten ein:

RMZ

Zins B3/12 = 0,00125

Zzr B6 = 36

Bw 0 = 0

Zw B1 = 10000

F 1 = 1

= -271,4089357

Liefert die konstante Zahlung einer Annuität pro Periode.

F kann den Wert 0 oder 1 annehmen und gibt an, wann Zahlungen fällig sind (Fälligkeit).

Formelergbnis = -271,41 €

OK Abbrechen

Abbildung 49

neben den oben beschriebenen Angaben geben Sie hier noch den Zukunftswert (Zw) und die Fälligkeit (F) an. dabei sei angenommen, dass der

- Bw** (Barwert zum jetzigen Zeitpunkt) gleich Null ist, es ist ja noch kein Geld da. Dagegen betrage der
- Zw** Zukunftswert = 10.000 €. Diesen Betrag möchten Sie in 3 Jahren angespart haben.
- F** Fälligkeit = 1 bedeutet, dass Sie Ihre Zahlungen am Monatsersten leisten. Dann bringt das Geld für den laufenden Monat noch Zinsen und die Rate wird dadurch kleiner. Alternativ können Sie am Monatsletzten einzahlen (F = 0 oder einfach freilassen). Dann wird der Zahlbetrag etwas größer.

7.2 Die Funktionen ZINS, Zw, Bw, ZZR

Alle bei der Funktion RMZ beschriebenen Angaben kann man in Excel auch mit Hilfe entsprechender Funktionen berechnen. Sie müssen nur die anderen Angaben kennen.

So lässt sich der ZINS **eines Ratenkredites** berechnen, wenn man den Betrag der Rate (Rmz), den Barwert (Kreditbetrag) und die Anzahl der Zahlungen (Zzr) kennt. Sehen Sie dazu als Beispiel die Werte aus der Abbildung 47. Wenn dort alles richtig berechnet war, müsste als Ergebnis ja 5,0 % herauskommen.

	A	B	C
1	Kreditsumme:	10.000 €	
2			
3	Zinssatz:		
4			
5	Laufzeit		5 Jahre
6			60 Monate
7			
8	Rate	188,71 €	

Abbildung 50

Nun benötigen Sie in Zelle B3 die Funktion ZINS. Wie bei der ersten Aufgabe zur Funktion RMZ gehen wir davon aus, dass der Barwert zunächst negativ ist. Am Ende des Zahlungszeitraumes

soll die Schuld getilgt sein (Zw = 0 oder leer). Gezahlt (Besser gesagt: gutgeschrieben) wird die Rate am Monatsende, also F = 0 (oder leer). Auf den Schätzwert, den es als sechstes Argument dieser Funktion noch gibt, verzichten Sie. Der ist in der Regel nicht notwendig und soll nur die Rechenzeit bei aufwendigen Berechnungen beschleunigen und ein genaueres Ergebnis liefern, was aber fast nie notwendig ist.

ZINS

Zr B6 = 60

Rmz B8 = 188,71

Bw -B1 = -10000

Zw = Zahl

F = Zahl

= 0,004166242

Liefert den Zinssatz einer Annuität pro Periode.

Zw ist der zukünftige Wert (Endwert) oder der Kassenbestand, den Sie nach der letzten Zahlung erreicht haben möchten.

Formelergebnis = 0,42%

OK Abbrechen

Abbildung 51

Das Ergebnis (Sie sehen es unten auf der Abbildung 51) beträgt 0,42 %. !? Dies ist jedoch der Zins für eine Zahlungsperiode, also einen Monat! Den müssen Sie noch mit 12 multiplizieren. Die Formel in der Zelle B3 sieht dann so aus:

=ZINS(B6;B8;-B1)*12

Das Ergebnis stimmt dann: 5,0%.

Nächstes Beispiel. Sie zahlen am Jahresanfang regelmäßig auf Ihren Bausparvertrag 500 € ein, bei einem Zinssatz von 2 % soll der nach sieben Jahren zuteilungsreif sein. Dazu muss die Hälfte der Bausparsumme eingezahlt sein. Wie hoch sollte man die Bausparsumme wählen, um dies zu erreichen?

Sie wollen in diesem Fall den Zukunftswert (ZW) ermitteln. Geben Sie dazu folgende Tabelle ein. Das Endguthaben ist unser „Zukunftswert“, den müssen Sie noch mal verdoppeln, um auf die Bausparsumme zu kommen:

	A	B	C
1	Bausparsumme	=2*B3	
2			
3	Endguthaben		
4			
5	Zinssatz:	2,0%	
6			
7	Laufzeit	7 Jahre	
8			
9	Rate	500,00 €	
10			

Abbildung 52

In die Zelle B3 tragen Sie die Funktion Zw ein:

ZW

Zins B5 = 0,02
 Zzr B7 = 7
 Rmz -B9 = -500
 Bw = Zahl
 F 1 = 1

= 3791,484525

Liefert den zukünftigen Wert (Endwert) einer Investition.

F kann den Wert 0 oder 1 annehmen und gibt an, wann Zahlungen fällig sind (Fälligkeit).

Formelergbnis = 3.791,48 €

OK Abbrechen

Abbildung 53

In dem Falle ist die Rate negativ anzugeben, da Sie zu zahlen ist. Sonst erhalten Sie einen negativen Betrag als Ergebnis. F = 1, da Sie ja am Anfang des Jahres die Rate einzahlen. Mit dieser Berechnung sieht das Ergebnis dann so aus:

	A	B	C
1	Bausparsumme	7.582,97 €	
2			
3	Endguthaben	3.791,48 €	
4			
5	Zinssatz:	2,0%	
6			
7	Laufzeit	7 Jahre	
8			
9	Rate	500,00 €	
10			

Abbildung 54

Ganz ähnlich können Sie mit Hilfe der Funktionen BW den Barwert und ZZR die Anzahl der Zahlungen berechnen. Voraussetzung ist, dass die übrigen Angaben bekannt sind. Im Übrigen enthält das Programm Excel inzwischen zu fast allen Funktionen sehr gute und nachvollziehbare Beispiele. Rufen Sie einfach die Hilfe zu der entsprechenden Funktion auf!

Die folgenden Funktionen wurden in diesem Script beschrieben:

ABRUNDEN	8	POTENZ	5
ABS.....	5	PRODUKT	5
ANZAHL.....	10	QUOTIENT	6
ANZAHL2	10	RANG	12
ANZAHLLEEREZELLEN	10	REST	6
AUFRUNDEN	8	RMZ	32
BW	34	RÖMISCH.....	5
COS	6	RUNDEN	7
DATUM.....	14	SEKUNDE	14
FAKULTÄT	5	SIN	6
GANZZAHL.....	5	STUNDE.....	14
GERADE	8	SUMME	2, 4
GGT	6	SUMMEWENN	23
HEUTE	14	SVERWEIS	24
INDEX	30, 32	TAG	14
JAHR	14	TAGE360	14
JETZT	14	TAN	6
KGRÖSSTE.....	11	UND	22
KGV	6	UNGERADE.....	8
KKLEINSTE	12	UNTERGRENZE	8
KOMBINATIONEN.....	6	VERGLEICH.....	30, 32
KÜRZEN.....	8	VERWEIS	28
LOG10	5	WENN.....	15
MAX	9	WENN (VERSCHACHELTELT).....	17
MIN	10	WURZEL	5
MINUTE.....	14	WVERWEIS.....	27
MITTELWERT.....	10	ZÄHLENWENN	23
MONAT	14	ZEIT	14
NICHT	22	ZINS.....	34
OBERGRENZE	8	ZUFALLSZAHL	6
ODER.....	21	ZW.....	35
PI.....	6	ZZR.....	34